

# Data Strategies in VFP: Advanced

By Doug Hennig

## Introduction

In the “Data Strategies in VFP: Introduction” document, we examined the different mechanisms for accessing non-VFP data (such as SQL Server) in VFP applications: remote views, SQL passthrough, ADO, XML, and the CursorAdapter class added in VFP 8. In this document, we’ll look at CursorAdapter in a lot more detail and discuss the concept of reusable data classes. In addition, we’ll take a brief look at the new XMLAdapter base class and see how it can help exchange data with other sources, such as ADO.NET.

## CursorAdapter

In my opinion, CursorAdapter is one of the biggest new features in VFP 8. The reasons I think they’re so cool are:

- They make it easy to use ODBC, ADO, or XML, even if you’re not very familiar with these technologies.
- They provide a consistent interface to remote data regardless of the mechanism you choose.
- They make it easy to switch from one mechanism to another.

Here’s an example of the last point. Suppose you have an application that uses ODBC with CursorAdapters to access SQL Server data, and for some reason you want to change to use ADO instead. All you need to do is change the DataSourceType of the CursorAdapters and change the connection to the backend database, and you’re done. The rest of the components in the application neither know nor care about this; they still see the same cursor regardless of the mechanism used to access the data.

Let’s start examining CursorAdapters by looking at their properties, events, and methods (PEMS).

## PEMS

We won’t discuss all of the properties, events, and methods of the CursorAdapter class here, just the more important ones. See the VFP documentation for the complete list.

## DataSourceType

This property is a biggie: it determines the behavior of the class and what kinds of values to put into some of the other properties. The valid choices are “Native”, which indicates that you’re using native tables, or “ODBC”, “ADO”, or “XML”, which means you’re using the appropriate

mechanism to access the data. You likely won't use "Native", since you would probably use a Cursor object rather than CursorAdapter, but this setting would make it easier to later upsize an application.

## **DataSource**

This is the means to access the data. VFP ignores this property when DataSourceType is set to "Native" or "XML". For ODBC, set DataSource to a valid ODBC connection handle (meaning you have to manage the connection yourself). In the case of ADO, DataSource must be an ADO RecordSet that has its ActiveConnection object set to an open ADO Connection object (again, you have to manage this yourself).

## **UseDEDataSource**

If this property is set to .T. (the default is .F.), you can leave the DataSourceType and DataSource properties alone since the CursorAdapter will use the DataEnvironment's properties instead (VFP 8 adds DataSourceType and DataSource to the DataEnvironment class as well). An example of when you'd set this to .T. is when you want all the CursorAdapters in a DataEnvironment to use the same ODBC connection.

## **SelectCmd**

In the case of everything but XML, this is the SQL SELECT command used to retrieve the data. In the case of XML, this can either be a valid XML string that can be converted into a cursor (using an internal XMLTOCURSOR() call) or an expression (such as a UDF) that returns a valid XML string.

## **CursorSchema**

This property holds the structure of the cursor in the same format as you'd use in a CREATE CURSOR command (everything between the parentheses in such a command). Here's an example: CUST\_ID C(6), COMPANY C(30), CONTACT C(30), CITY C(25). Although it's possible to leave this blank and tell the CursorAdapter to determine the structure when it creates the cursor, it works better if you fill CursorSchema in. For one thing, if CursorSchema is blank or incorrect, you'll either get errors when you open the DataEnvironment of a form or you won't be able to drag and drop fields from the CursorAdapter to the form to create controls. Fortunately, the CursorAdapter Builder that comes with VFP can automatically fill this in for you.

## **AllowDelete, AllowInsert, AllowUpdate, and SendUpdates**

These properties, which default to .T., determine if deletes, inserts, and updates can be done and if changes are sent to the data source.

## **KeyFieldList, Tables, UpdatableFieldList, and UpdateNameList**

These properties, which serve the same purpose as the same-named CURSORSETPROP() properties for views, are required if you want VFP to automatically update the data source with

changes made in the cursor. KeyFieldList is a comma-delimited list of fields (without aliases) that make up the primary key for the cursor. Tables is a comma-delimited list of tables. UpdatableFieldList is a comma-delimited list of fields (without aliases) that can be updated. UpdateNameList is a comma-delimited list that matches field names in the cursor to field names in the table. The format for UpdateNameList is as follows: CURSORFIELDNAME1 TABLE.FIELDNAME1, CURSORFIELDNAME2 TABLE.FIELDNAME2, ... Note that even if UpdatableFieldList doesn't contain the name of the primary key of the table (because you don't want that field updated), it must still exist in UpdateNameList or updates won't work.

### **\*Cmd, \*CmdDataSource, \*CmdDataSourceType**

If you want to specifically control how VFP deletes, inserts, or updates records in the data source, you can assign the appropriate values to these sets of properties (replace the \* above with Delete, Insert, and Update).

### **CursorFill(UseCursorSchema, NoData, Options, Source)**

This method creates the cursor and fills it with data from the data source (although you can pass .T. for the NoData parameter to create an empty cursor). Pass .T. for the first parameter to use the schema defined in CursorSchema or .F. to create an appropriate structure from the data source (in my opinion, this behavior is reversed). MULTILOCKS must be set on or this method will fail. If CursorFill fails for any reason, it returns .F. rather than raising an error; use AERROR() to determine what went wrong (although be prepared for some digging, since the error messages you get often aren't specific enough to tell you exactly what the problem is).

### **CursorRefresh()**

This method is similar to the REQUERY() function: it refreshes the cursor's contents.

### **Before\*() and After\*()**

Pretty much every method and event has before and after "hook" events that allow you to customize the behavior of the CursorAdapter. For example, in AfterCursorFill, you can create indexes for the cursor so they're always available. In the case of the Before events, you can return .F. to prevent the action that trigger it from occurring (this is similar to database events).

Here's an example (CursorAdapterExample.prg) that gets certain fields for Brazilian customers from the Customers table in the Northwind database that comes with SQL Server. The cursor is updateable, so if you make changes in the cursor, close it, and then run the program again, you'll see that your changes were saved to the backend.

```
local loCursor as CursorAdapter, ;
    laErrors[1]
loCursor = createobject('CursorAdapter')
with loCursor
    .Alias                = 'Customers'
    .DataSourceType      = 'ODBC'
    .DataSource          = sqlstringconnect('driver=SQL Server;' + ;
        'server=(local);database=Northwind;uid=sa;pwd=;trusted_connection=no')
```

```

.SelectCmd      = "select CUSTOMERID, COMPANYNAME, CONTACTNAME " + ;
  "from CUSTOMERS where COUNTRY = 'Brazil'"
.KeyFieldList  = 'CUSTOMERID'
.Tables        = 'CUSTOMERS'
.UpdatableFieldList = 'CUSTOMERID, COMPANYNAME, CONTACTNAME'
.UpdateNameList = 'CUSTOMERID CUSTOMERS.CUSTOMERID, ' + ;
  'COMPANYNAME CUSTOMERS.COMPANYNAME, CONTACTNAME CUSTOMERS.CONTACTNAME'
if .CursorFill()
  browse
else
  aerror(laErrors)
  messagebox(laErrors[2])
endif .CursorFill()
endwith

```

## DataEnvironment and Form Changes

To support the new CursorAdapter class, several changes have been made to the DataEnvironment and Form classes and their designers.

First, as I mentioned earlier, the DataEnvironment class now has DataSource and DataSourceType properties. It doesn't use these properties itself but they're used by any CursorAdapter member that has UseDEDataSource set to .T. Second, you can now create DataEnvironment subclasses visually using the Class Designer (woo hoo!).

As for forms, you can now specify a DataEnvironment subclass to use by setting the new DEClass and DEClassLibrary properties. If you do this, anything you've done with the existing DataEnvironment (cursors, code, etc.) will be lost, but at least you're warned first. A cool new feature of forms that's somewhat related is the BindControls property; setting this to .F. in the Property Window means that VFP won't try to data bind the controls at init time, only when you set BindControls to .T. What's this good for? Well, how many times have you cursed that parameters are passed to Init, which fires after all the controls have been initialized and bound to their ControlSources? What if you want to pass a parameter to a form that tells it which table to open or other things that affect the ControlSources? This new property makes this issue a snap.


## Other Changes

CURSORGETPROP('SourceType') returns a new range of values: if the cursor was created with CursorFill, the value is 100 + the old value (for example, 102 for remote data). If the cursor was created with CursorAttach (which allows you to attach an existing cursor to a CursorAdapter object), the value is 200 + the old value. If the data source is an ADO RecordSet, the value is 104 (CursorFill) or 204 (CursorAttach).


## Builders

VFP includes DataEnvironment and CursorAdapter builders that make it easier to work with these classes.


The DataEnvironment Builder is brought up in the usual way: by right-clicking on the DataEnvironment of a form or on a DataEnvironment subclass in the Class Designer and choosing Builder. The “Data Source” page of the DataEnvironment Builder is where you set data source information. Choose the desired data source type and where the data source comes from. If you choose “Use existing connection handle” (ODBC) or “Use existing ADO RecordSet” (ADO), specify the expression containing the data source (such as “goConnectionMgr.nHandle”). You can also choose to use one of the DSNs on your system or a connection string. The Build button, which is only enabled if you choose “Use connection string” for ADO, displays the Data Link Properties dialog, which you can use to build the connection string visually. If you select either “Use DSN” or “Use connection string”, the builder will generate code in the BeforeOpenTables method of the DataEnvironment to create the desired connection. If you choose “Native”, you can select a VFP database container as a data source; in that case, the generated code will ensure the database is open (you can also use free tables as the data source).




The “Cursors” page allows you to maintain the CursorAdapter members of the DataEnvironment (Cursor objects don’t show up, nor can they be added, in the builder). The Add button allows you to add a CursorAdapter subclass to the DataEnvironment, while New create a new base class CursorAdapter. Remove deletes the select CursorAdapter and Builder invokes the CursorAdapter Builder for the selected CursorAdapter. You can change the name of the CursorAdapter object, but you’ll need the CursorAdapter Builder for any other properties.




The CursorAdapter Builder is also invoked by choosing Builder from the shortcut menu. The “Properties” page shows the class and name of the object (Name can only be changed if the builder is brought up from a DataEnvironment, since it’s read-only for a CursorAdapter subclass), the alias of the cursor it’ll create, whether the DataEnvironment’s data source should be used or not, and connection information if not. As with the DataEnvironment Builder, the CursorAdapter Builder will generate code to create the desired connection (in the CursorFill method in this case) if you select either “Use DSN” or “Use connection string”.



The “Data Access” page allows you to specify the SelectCmd, CursorSchema, and other properties. If you have specified connection information, you can click on the Build button for SelectCmd to display the Select Command Builder, which makes it easy to create the SelectCmd.




The Select Command Builder makes short work of building a simple SELECT statement. Choose the desired table from the table dropdown, then move the appropriate fields to the selected side. In the case of a native data source, you can add tables to the Table combobox (such as if you want use free tables). When you choose OK, the SelectCmd will be filled with the appropriate SQL SELECT statement.




Click on the Build button for the CursorSchema to have this property filled in for you automatically. In order for this to work, the builder actually creates a new CursorAdapter object, sets the properties appropriately, and calls CursorFill to create the cursor. If you don't have a live connection to the data source or CursorFill fails for some reason (such as an invalid SelectCmd), this obviously won't work.



Use the Auto-Update page to set the properties necessary for VFP to automatically generate update statements for the data source. The Tables property is automatically filled in from the tables specified in SelectCmd, and the fields grid is filled in from the fields in CursorSchema. Like the View Designer, you select which are the key fields and which fields are updatable by checking the appropriate column in the grid. You can also set other properties, such as functions to convert the data in certain fields of the cursor before sending it to the data source.



The Update, Insert, and Delete pages have a nearly identical appearance. They allow you to specify values for the sets of Update, Delete, and Insert properties. This is especially important for XML, for which VFP can't automatically generate update statements.



## Using Native Data

Even though it's clear that CursorAdapter was intended to standardize and simplify the access to non-VFP data, you can use it as a substitute for Cursor by setting DataSourceType to "Native". Why would you do this? Mostly as a look toward the future when your application might be upsized; by simply changing the DataSourceType to one of the other choices (and likely changing a few other properties such as setting connection information), you can easily switch to another DBMS such as SQL Server.

When DataSourceType is set to "Native", VFP ignores DataSource. SelectCmd must be a SQL SELECT statement, not a USE command or expression, so that means you're always working with the equivalent of a local view rather than the table directly. You're responsible for ensuring that VFP can find any tables referenced in the SELECT statement, so if the tables aren't in the current directory, you either need to set a path or open the database the tables belong to. As usual, if you want the cursor to be updateable, be sure to set the update properties (KeyFieldList, Tables, UpdatableFieldList, and UpdateNameList).

The following example (NativeExample.prg) creates an updateable cursor from the Customer table in the TestData VFP sample database:

```
local loCursor as CursorAdapter, ;
    laErrors[1]
open database (_samples + 'data\testdata')
```

```

loCursor = createobject('CursorAdapter')
with loCursor
    .Alias          = 'customercursor'
    .DataSourceType = 'Native'
    .SelectCmd      = "select CUST_ID, COMPANY, CONTACT from CUSTOMER " + ;
        "where COUNTRY = 'Brazil'"
    .KeyFieldList   = 'CUST_ID'
    .Tables         = 'CUSTOMER'
    .UpdatableFieldList = 'CUST_ID, COMPANY, CONTACT'
    .UpdateNameList = 'CUST_ID CUSTOMER.CUST_ID, ' + ;
        'COMPANY CUSTOMER.COMPANY, CONTACT CUSTOMER.CONTACT'
    if .CursorFill()
        browse
        tableupdate(1)
    else
        aerror(laErrors)
        messagebox(laErrors[2])
    endif .CursorFill()
endwith
close databases all

```

## Using ODBC

ODBC is actually the most straightforward of the four settings of DataSourceType. You set DataSource to an open ODBC connection handle, set the usual properties, and call CursorFill to retrieve the data. If you fill in KeyFieldList, Tables, UpdatableFieldList, and UpdateNameList, VFP will automatically generate the appropriate UPDATE, INSERT, and DELETE statements to update the backend with any changes. If you want to use a stored procedure instead, set the \*Cmd, \*CmdDataSource, and \*CmdDataSourceType properties appropriately.

Here's an example, taken from ODBCExample.prg, that calls the CustOrderHist stored procedure in the Northwind database to get total units sold by product for a specific customer:

```

local loCursor as CursorAdapter, ;
    laErrors[1]
loCursor = createobject('CursorAdapter')
with loCursor
    .Alias          = 'CustomerHistory'
    .DataSourceType = 'ODBC'
    .DataSource     = sqlstringconnect('driver=SQL Server;server=(local);' + ;
        'database=Northwind;uid=sa;pwd=;trusted_connection=no')
    .SelectCmd      = "exec CustOrderHist 'ALFKI'"
    if .CursorFill()
        browse
    else
        aerror(laErrors)
        messagebox(laErrors[2])
    endif .CursorFill()
endwith

```

## Using ADO

Using ADO as the data access mechanism with CursorAdapter has a few more issues than using ODBC:

- DataSource must be set to an ADO RecordSet that has its ActiveConnection property set to an open ADO Connection object.
- If you want to use a parameterized query (which is likely the usual case rather than retrieving all records), you have to pass an ADO Command object that has its ActiveConnection property set to an open ADO Connection object as the fourth parameter to CursorFill. VFP will take care of filling the Parameters collection of the Command object (it parses SelectCmd to find the parameters) for you, but of course the variables containing the values of the parameters must be in scope.
- Using one CursorAdapter with ADO in a DataEnvironment is straightforward: you can set UseDEDataSource to .T. if you wish, then set the DataEnvironment's DataSource and DataSourceType properties as you would with the CursorAdapter. However, this doesn't work if there's more than one CursorAdapter in the DataEnvironment. The reason is that the ADO RecordSet referenced by DataEnvironment.DataSource can only contain a single CursorAdapter's data; when you call CursorFill for the second CursorAdapter, you get a "RecordSet is already open" error. So, if your DataEnvironment has more than one CursorAdapter, you must set UseDEDataSource to .F and manage the DataSource and DataSourceType properties of each CursorAdapter yourself (or perhaps use a DataEnvironment subclass that manages that for you).

The sample code below, taken from ADOExample.prg, shows how to retrieve data using a parameterized query with the help of an ADO Command object. This example also shows the use of the new structured error handling features in VFP 8; the call to the ADO Connection Open method is wrapped in a TRY ... CATCH ... ENDTRY statement to trap the COM error the method will throw if it fails.

```
local loConn as ADODB.Connection, ;
    loCommand as ADODB.Command, ;
    loException as Exception, ;
    loCursor as CursorAdapter, ;
    lcCountry, ;
    laErrors[1]
loConn = createobject('ADODB.Connection')
with loConn
    .ConnectionString = 'provider=SQLOLEDB.1;data source=(local);' + ;
        'initial catalog=Northwind;uid=sa;pwd=;trusted_connection=no'
    try
        .Open()
    catch to loException
        messagebox(loException.Message)
        cancel
    endtry
endwith
loCommand = createobject('ADODB.Command')
```

```

loCursor = createobject('CursorAdapter')
with loCursor
  .Alias          = 'Customers'
  .DataSourceType = 'ADO'
  .DataSource     = createobject('ADODB.RecordSet')
  .SelectCmd     = 'select * from customers where country=?lcCountry'
  lcCountry      = 'Brazil'
  .DataSource.ActiveConnection = loConn
  loCommand.ActiveConnection   = loConn
  if .CursorFill(.F., .F., 0, loCommand)
    browse
  else
    aerror(laErrors)
    messagebox(laErrors[2])
  endif .CursorFill(.F., .F., 0, loCommand)
endwith

```

## Using XML

Using XML with CursorAdapters requires some additional things. Here are the issues:

- The DataSource property is ignored.
- The CursorSchema property must be filled in, even if you pass .F. as the first parameter to the CursorFill method, or you'll get an error.
- The SelectCmd property must be set to an expression, such as a user-defined function (UDF) or object method name, that returns the XML for the cursor.
- Changes made to the cursor are converted to a diffgram, which is XML that contains before and after values for changed fields and records, and placed in the UpdateGram property when the update is required.
- In order to write changes back to the data source, UpdateCmdDataSourceType must be set to "XML" and UpdateCmd must be set to an expression (again, likely a UDF or object method) that handles the update. You'll probably want to pass "This.UpdateGram" to the UDF so it can send the changes to the data source.

The XML source for the cursor could come from a variety of places. For example, you could call a UDF that converts a VFP cursor into XML using CURSORTOXML() and returns the results:

```

use CUSTOMERS
cursortoxml('customers', 'lcXML', 1, 8, 0, '1')
return lcXML

```

The UDF could call a Web Service that returns a result set as XML. Here's an example that IntelliSense generated for me from a Web Service I created and registered on my own system (the details aren't important; it just shows an example of a Web Service).

```

local loWS as dataserver web service
loWS = NEWOBJECT("Wsclient", HOME()+"ffc\_webservices.vcx")

```

```

loWS.cWSName = "dataserver web service"
loWS = loWS.SetupClient("http://localhost/SQDataServer/dataserver.WSDL", ;
    "dataserver", "dataserverSoapPort")
lcXML = loWS.GetCustomers()
return lcXML

```

It could use SQLXML 3.0 to execute a SQL Server 2000 query stored in a template file on a Web Server (for more information on SQLXML, go to <http://msdn.microsoft.com> and search for SQLXML). The following code uses an MSXML2.XMLHTTP object to get all records from the Northwind Customers table via HTTP; this will be explained in more detail later.

```

local loXML as MSXML2.XMLHTTP
loXML = createobject('MSXML2.XMLHTTP')
loXML.open('POST', 'http://localhost/northwind/template/' + ;
    'getallcustomers.xml, .F.)
loXML.setRequestHeader('Content-type', 'text/xml')
loXML.send()
return loXML.responseText

```

Handling updates is more complicated. The data source must either be capable of accepting and consuming a diffgram (as is the case with SQL Server 2000) or you'll have to figure out the changes yourself and issue a series of SQL statements (UPDATE, INSERT, and DELETE) to perform the updates.

Here's an example (XMLExample.prg) that uses a CursorAdapter with an XML data source. Notice that both SelectCmd and UpdateCmd call UDFs. In the case of SelectCmd, the name of a SQL Server 2000 XML template and the customer ID to retrieve is passed to a UDF called GetNWXML, which we'll look at in a moment. For UpdateCmd, VFP passed the UpdateGram property to SendNWXML, which we'll also look at later.

```

local loCustomers as CursorAdapter, ;
    laErrors[1]
loCustomers = createobject('CursorAdapter')
with loCustomers
    .Alias = 'Customers'
    .CursorSchema = 'CUSTOMERID C(5), COMPANYNAME C(40), ' + ;
        'CONTACTNAME C(30), CONTACTTITLE C(30), ADDRESS C(60), ' + ;
        'CITY C(15), REGION C(15), POSTALCODE C(10), COUNTRY C(15), ' + ;
        'PHONE C(24), FAX C(24) '
    .DataSourceType = 'XML'
    .KeyFieldList = 'CUSTOMERID'
    .SelectCmd = 'GetNWXML([customersbyid.xml?customerid=ALFKI]) '
    .Tables = 'CUSTOMERS'
    .UpdatableFieldList = 'CUSTOMERID, COMPANYNAME, CONTACTNAME, ' + ;
        'CONTACTTITLE, ADDRESS, CITY, REGION, POSTALCODE, COUNTRY, PHONE, FAX '
    .UpdateCmdDataSourceType = 'XML'
    .UpdateCmd = 'SendNWXML(This.UpdateGram) '
    .UpdateNameList = 'CUSTOMERID CUSTOMERS.CUSTOMERID, ' + ;
        'COMPANYNAME CUSTOMERS.COMPANYNAME, ' + ;
        'CONTACTNAME CUSTOMERS.CONTACTNAME, ' + ;
        'CONTACTTITLE CUSTOMERS.CONTACTTITLE, ' + ;
        'ADDRESS CUSTOMERS.ADDRESS, ' + ;
        'CITY CUSTOMERS.CITY, ' + ;
        'REGION CUSTOMERS.REGION, ' + ;

```

```

        'POSTALCODE CUSTOMERS.POSTALCODE, ' + ;
        'COUNTRY CUSTOMERS.COUNTRY, ' + ;
        'PHONE CUSTOMERS.PHONE, ' + ;
        'FAX CUSTOMERS.FAX'
    if .CursorFill(.T.)
        browse
    else
        aerror(laErrors)
        messagebox(laErrors[2])
    endif .CursorFill(.T.)
endwith

```

The XML template this code references, CustomersByID.XML, looks like the following:

```

<root xmlns:sql="urn:schemas-microsoft-com:xml-sql">
  <sql:header>
    <sql:param name="customerid">
      </sql:param>
    </sql:header>
    <sql:query client-side-xml="0">
      SELECT *
      FROM Customers
      WHERE CustomerID = @customerid
      FOR XML AUTO
    </sql:query>
  </root>

```

Place this file in a virtual directory for the Northwind database (see the appendix for details on configuring IIS to work with SQL Server).

Here's the code for GetNWXML. It uses an MSXML2.XMLHTTP object to access a SQL Server 2000 XML template on a Web server and returns the results. The name of the template (and optionally any query parameters) is passed as a parameter to this code.

```

lparameters tcURL
local loXML as MSXML2.XMLHTTP
loXML = createobject('MSXML2.XMLHTTP')
loXML.open('POST', 'http://localhost/northwind/template/' + tcURL, .F.)
loXML.setRequestHeader('Content-type', 'text/xml')
loXML.send()
return loXML.responseText

```

SendNWXML looks similar, except it expects to be passed a diffgram, loads the diffgram into an MSXML2.DOMDocument object, and passes that object to the Web Server, which will in turn pass it via SQLXML to SQL Server 2000 for processing.

```

lparameters tcDiffGram
local loDOM as MSXML2.DOMDocument, ;
  loXML as MSXML2.XMLHTTP
loDOM = createobject('MSXML2.DOMDocument')
loDOM.async = .F.
loDOM.loadXML(tcDiffGram)
loXML = createobject('MSXML2.XMLHTTP')
loXML.open('POST', 'http://localhost/northwind/', .F.)

```

```
loXML.setRequestHeader('Content-type', 'text/xml')
loXML.send(loDOM)
```

To see how this works, run XMLExample.prg. You should see a single record (the ALFKI customer) in a browse window. Change the value in some field, then close the window and run the PRG again. You should see that your change was written to the backend.

## CursorAdapter and DataEnvironment Subclasses

As is normally the case in VFP, I've created subclasses of CursorAdapter and DataEnvironment that I'll use rather than the base classes.

### SFCursorAdapter

SFCursorAdapter (in SFDataClasses.vcx) is a subclass of CursorAdapter that has some additional functionality added:

- It can automatically handle parameterized queries; you can define the parameter values as static (a constant value) or dynamic (an expression, such as “=Thisform.txtName.Value”, that's evaluated when the cursor is opened or refreshed).
- It can automatically create indexes on the cursor after it's opened.
- It does some special things for ADO, such as setting DataSource to an ADO RecordSet, setting the ActiveConnection property of the RecordSet to an ADO Connection object, and creating and passing an ADO Command object to CursorFill when a parameterized query is used.
- It provides some simple error handling (the cErrorMessage property is filled with the error message).
- It has Update and Release methods that are missing in CursorAdapter.

Let's take a look at this class.

The Init method creates two collections (using the new Collection base class, which maintains collections of things), one for parameters that may be needed for the SelectCmd property, and one for tags that should be created automatically after the cursor is opened. It also sets MULTILOCKS on, since that's required for CursorAdapter cursors.

```
with This
* Create parameters and tags collections.

  .oParameters = createobject('Collection')
  .oTags       = createobject('Collection')

* Ensure MULTILOCKS is set on.
```



```
set multilocks on
endwith
```

The `AddParameter` method adds a parameter to the parameters collection. Pass this method the name of the parameter (this should match the name as it appears in the `SelectCmd` property) and optionally the value of the parameter (if you don't pass it now, you can set it later by using the `GetParameter` method). This code shows a couple of new features in VFP 8: the new `Empty` base class, which has no PEMs, making it ideal for lightweight objects, and the `ADDPROPERTY()` function, which acts like the `AddProperty` method for those objects that don't have that method.

```
lparameters tcName, ;
    tuValue
local loParameter
loParameter = createobject('Empty')
addproperty(loParameter, 'Name', tcName)
addproperty(loParameter, 'Value', tuValue)
This.oParameters.Add(loParameter, tcName)
```

Use the `GetParameter` method to return a specific parameter object; this is normally used when you want to set the value to use for the parameter.

```
lparameters tcName
local loParameter
loParameter = This.oParameters.Item(tcName)
return loParameter
```

The `SetConnection` method is used to set the `DataSource` property to the desired connection. If `DataSourceType` is "ODBC", pass the connection handle. If it's "ADO", `DataSource` needs to be an ADO `RecordSet` with its `ActiveConnection` property set to an open ADO `Connection` object, so pass the `Connection` object and `SetConnection` will create the `RecordSet` and set its `ActiveConnection` to the passed object.

```
lparameters tuConnection
with This
do case
case .DataSourceType = 'ODBC'
.DataSource = tuConnection
case .DataSourceType = 'ADO'
.DataSource = createobject('ADODB.RecordSet')
.DataSource.ActiveConnection = tuConnection
endcase
endwith
```

To create the cursor, call the `GetData` method rather than `CursorFill`, since it handles parameters and errors automatically. Pass `.T.` to `GetData` if you want the cursor created but not filled with data. The first thing this method does is create privately-scoped variables with the same names and values as the parameters defined in the parameters collection (the `GetParameterValue` method called from here returns either the `Value` of the parameter object or the evaluation of the `Value` if it starts with an "="). Next, if we're using ADO and there are any parameters, the code creates an ADO `Command` object and sets its `ActiveConnection` to the `Connection` object, then passes the `Command` object to the `CursorFill` method; the `CursorAdapter` requires this for

parameterized ADO queries. If we're not using ADO or we don't have any parameters, the code simply calls CursorFill to fill the cursor. Note that .T. is passed to CursorFill, telling it to use CursorSchema, if CursorSchema is filled in (this is the behavior I wish the base class had). If the cursor was created, the code calls the CreateTags method to create the desired indexes for the cursor; if not, it calls the HandleError method to handle any errors that occurred.

```
lparameters tlNoData
local loParameter, ;
    lcName, ;
    luValue, ;
    llUseSchema, ;
    loCommand, ;
    llReturn
with This

* If we're supposed to fill the cursor (as opposed to creating an empty one),
* create variables to hold any parameters. We have to do it here rather than
* in a method since we want them to be scoped as private.

    if not tlNoData
        for each loParameter in .oParameters
            lcName = loParameter.Name
            luValue = .GetParameterValue(loParameter)
            store luValue to (lcName)
        next loParameter
    endif not tlNoData

* If we're using ADO and there are any parameters, we need a Command object
* to handle this.

    llUseSchema = not empty(.CursorSchema)
    if '?' $ .SelectCmd and (.DataSourceType = 'ADO' or (.UseDEDataSource and ;
        .Parent.DataSourceType = 'ADO'))
        loCommand = createobject('ADODB.Command')
        loCommand.ActiveConnection = iif(.UseDEDataSource, ;
            .Parent.DataSource.ActiveConnection, .DataSource.ActiveConnection)
        llReturn = .CursorFill(llUseSchema, tlNoData, .nOptions, loCommand)
    else

* Try to fill the cursor.

        llReturn = .CursorFill(llUseSchema, tlNoData, .nOptions)
    endif '?' $ .SelectCmd ...

* If we created the cursor, create any tags defined for it. If not, handle
* the error.

    if llReturn
        .CreateTags()
    else
        .HandleError()
    endif llReturn
endwith
return llReturn
```

The Requery method is similar to GetData, except that it refreshes the data in the cursor. Call this method rather than CursorRefresh because, as with GetData, it handles parameters and errors.

```
local loParameter, ;
    lcName, ;
    luValue, ;
    llReturn
with This

* Create variables to hold any parameters. We have to do it here rather than
* in a method since we want them to be scoped as private.

for each loParameter in .oParameters
    lcName = loParameter.Name
    luValue = .GetParameterValue(loParameter)
    store luValue to (lcName)
next loParameter

* Refresh the cursor; handle any error if it failed.

llReturn = .CursorRefresh()
if not llReturn
    .HandleError()
endif not llReturn
endwith
return llReturn
```

The Update method is simple: it simply calls TABLEUPDATE() to attempt to update the original data source and calls HandleError if it failed.

```
local llReturn
llReturn = tableupdate(1, .F., This.Alias)
if not llReturn
    This.HandleError()
endif not llReturn
return llReturn
```

There are a few methods we won't look at here; feel free to examine them yourself. AddTag adds information about an index you want created after the cursor is created to the tags collection, while CreateTags, which is called from GetData, uses the information in that collection in INDEX ON statements. HandleError uses AERROR() to determine what went wrong and puts the second element of the error array into the cErrorMessage property.

Let's look at a couple of examples of using this class. The first one (taken from TestCursorAdapter.prg) gets all records from the Customers table in the Northwind database. This code isn't all that different from what you'd use for a base class CursorAdapter (you'd have to pass .F. for the first parameter to CursorFill since the CursorSchema isn't filled in).

```
loCursor = newobject('SFCursorAdapter', 'SFDataClasses')
with loCursor

* Connect to the SQL Server Northwind database and get customers records.
```

```

.DataSourceType = 'ODBC'
.DataSource     = sqlstringconnect('driver=SQL Server;server=(local);' + ;
    'database=Northwind;uid=sa;pwd=;trusted_connection=no')
.Alias         = 'Customers'
.SelectCommand  = 'select * from customers'
if .GetData()
    browse
else
    messagebox('Could not get the data. The error message was:' + ;
        chr(13) + chr(13) + .cErrorMessage)
endif .GetData()
endwith

```

The next example (also taken from TestCursorAdapter.prg) uses the ODBC version of the SFConnectionMgr, which we looked at in the “Data Strategies in VFP: Introduction” document, to manage the connection. It also uses a parameterized statement for SelectCmd, shows how the AddParameter method allows you to handle parameters, and demonstrates how you can automatically create tags for the cursor using the AddTag method.

```

loConnMgr = newobject('SFConnectionMgrODBC', 'SFRemote')
with loConnMgr
    .cDriver   = 'SQL Server'
    .cServer   = '(local)'
    .cDatabase = 'Northwind'
    .cUserName = 'sa'
    .cPassword = ''
endwith
if loConnMgr.Connect()
    loCursor = newobject('SFCursorAdapter', 'SFDataClasses')
    with loCursor
        .DataSourceType = 'ODBC'
        .SetConnection(loConnMgr.GetConnection())
        .Alias           = 'Customers'
        .SelectCmd      = 'select * from customers where country = ?pcountry'
        .AddParameter('pcountry', 'Brazil')
        .AddTag('CustomerID', 'CustomerID')
        .AddTag('Company',    'upper(CompanyName)')
        .AddTag('Contact',    'upper(ContactName)')
        if .GetData()
            messagebox('Brazilian customers in CustomerID order')
            set order to CustomerID
            go top
            browse
            messagebox('Brazilian customers in Contact order')
            set order to Contact
            go top
            browse
            messagebox('Canadian customers')
            loParameter = .GetParameter('pcountry')
            loParameter.Value = 'Canada'
            .Requery()
            browse
        else
            messagebox('Could not get the data. The error message was:' + ;

```

```

        chr(13) + chr(13) + .cErrorMessage)
    endif .GetData()
endwith
else
    messagebox(loConnMgr.cErrorMessage)
endif loConnMgr.Connect()

```

## SFDataEnvironment

SFDataEnvironment (also in SFDataClasses.vcx) is much simpler than SFCursorAdapter, but has some useful functionality added:

- The GetData method calls the GetData method of all SFCursorAdapter members so you don't have to call them individually.
- Similarly, the Requery and Update methods call the Requery and Update methods of every SFCursorAdapter member.
- Like SFCursorAdapter, the SetConnection method sets DataSource to an ADO RecordSet and sets the ActiveConnection property of the RecordSet to an ADO Connection object. However, it also calls the SetConnection method of any SFCursorAdapter member that has UseDEDataSource set to .F.
- It provides some simple error handling (a cErrorMessage property is filled with the error message).
- It has a Release method.

GetData is very simple: it just calls the GetData method of any member object that has that method.

```

lparameters tlNoData
local loCursor, ;
llReturn
for each loCursor in This.Objects
    if pemstatus(loCursor, 'GetData', 5)
        llReturn = loCursor.GetData(tlNoData)
        if not llReturn
            This.cErrorMessage = loCursor.cErrorMessage
            exit
        endif not llReturn
    endif pemstatus(loCursor, 'GetData', 5)
next loCursor
return llReturn

```

SetConnection is a little more complicated: it calls the SetConnection method of any member object that has that method and that has UseDEDataSource set to .F., then uses code similar to that in SFCursorAdapter to set its own DataSource if any of the CursorAdapters have UseDEDataSource set to .T.

```

lparameters tuConnection
local llSetOurs, ;
    loCursor, ;
    llReturn
with This

* Call the SetConnection method of any CursorAdapter that isn't using our
* DataSource.

llSetOurs = .F.
for each loCursor in .Objects
do case
case upper(loCursor.BaseClass) <> 'CURSORADAPTER'
case loCursor.UseDEDataSource
    llSetOurs = .T.
case pemstatus(loCursor, 'SetConnection', 5)
    loCursor.SetConnection(tuConnection)
endcase
next loCursor

* If we found any CursorAdapters that are using our DataSource, we'll need to
* set our own.

if llSetOurs
do case
case .DataSourceType = 'ODBC'
    .DataSource = tuConnection
case .DataSourceType = 'ADO'
    .DataSource = createobject('ADODB.RecordSet')
    .DataSource.ActiveConnection = tuConnection
endcase
endif llSetOurs
endwith

```

Requery and Update are almost identical to GetData, so we won't bother looking at them.

TestDE.prg shows the use of SFDataEnvironment as a container for a couple of SFCursorAdapter classes. Since this example uses ADO, each SFCursorAdapter needs its own DataSource, so UseDEDataSource is set to .F. Notice that a single call to the DataEnvironment SetConnection method takes care of setting the DataSource property for each CursorAdapter.

```

loConnMgr = newobject('SFConnectionMgrADO', 'SFRemote')
with loConnMgr
    .cDriver    = 'SQLOLEDB.1'
    .cServer    = '(local)'
    .cDatabase  = 'Northwind'
    .cUserName  = 'sa'
    .cPassword  = ''
endwith
if loConnMgr.Connect()
loDE = newobject('SFDataEnvironment', 'SFDataClasses')
with loDE
    .NewObject('CustomersCursor', 'SFCursorAdapter', 'SFDataClasses')
with .CustomersCursor
    .Alias      = 'Customers'

```

```

        .SelectCmd      = 'select * from customers'
        .DataSourceType = 'ADO'
    endwhile
    .NewObject('OrdersCursor', 'SFCursorAdapter', 'SFDataClasses')
    with .OrdersCursor
        .Alias          = 'Orders'
        .SelectCmd      = 'select * from orders'
        .DataSourceType = 'ADO'
    endwhile
    .SetConnection(loConnMgr.GetConnection())
    if .GetData()
        select Customers
        browse nowait
        select Orders
        browse
    else
        messagebox('Could not get the data. The error message was:' + ;
            chr(13) + chr(13) + .ErrorMessage)
    endif .GetData()
endwith
else
    messagebox(loConnMgr.cErrorMessage)
endif loConnMgr.Connect()

```

## Reusable Data Classes

Now that we have CursorAdapter and DataEnvironment subclasses to work with, let's talk about reusable data classes.

One thing VFP developers have asked Microsoft to add to VFP for a long time is reusable data environments. For example, you may have a form and a report that have exactly the same data setup, but you have to manually fill in the DataEnvironment for each one because DataEnvironments aren't reusable. Some developers (and almost all frameworks vendors) made it easier to create reusable DataEnvironments by creating DataEnvironments in code (they couldn't be subclassed visually) and using a "loader" object on the form to instantiate the DataEnvironment subclass. However, this was kind of a kludge and didn't help with reports.

Now, in VFP 8, we have the ability to create both reusable data classes, which can provide cursors from any data source to anything that needs them, and reusable DataEnvironments, which can host the data classes. As of this writing, you can't use CursorAdapter or DataEnvironment subclasses in a report, but you can programmatically add CursorAdapter subclasses (such as in the Init method of the DataEnvironment) to take advantage of reusability there.

Let's create data classes for the Northwind Customers and Orders tables. First, create a subclass of SFCursorAdapter called CustomersCursor and set the properties as shown below.

Property	Value
----------	-------

Alias	Customers
CursorSchema	CUSTOMERID C(5), COMPANYNAME C(40), CONTACTNAME C(30), CONTACTTITLE C(30), ADDRESS C(60), CITY C(15), REGION C(15), POSTALCODE C(10), COUNTRY C(15), PHONE C(24), FAX C(24)
KeyFieldList	CUSTOMERID
SelectCmd	select * from customers
Tables	CUSTOMERS
UpdatableFieldList	CUSTOMERID, COMPANYNAME, CONTACTNAME, CONTACTTITLE, ADDRESS, CITY, REGION, POSTALCODE, COUNTRY, PHONE, FAX
UpdateNameList	CUSTOMERID CUSTOMERS.CUSTOMERID, COMPANYNAME CUSTOMERS.COMPANYNAME, CONTACTNAME CUSTOMERS.CONTACTNAME, CONTACTTITLE CUSTOMERS.CONTACTTITLE, ADDRESS CUSTOMERS.ADDRESS, CITY CUSTOMERS.CITY, REGION CUSTOMERS.REGION, POSTALCODE CUSTOMERS.POSTALCODE, COUNTRY CUSTOMERS.COUNTRY, PHONE CUSTOMERS.PHONE, FAX, CUSTOMERS.FAX

Note: you can use the CursorAdapter Builder to do most of the work, especially setting the CursorSchema and update properties. The trick is to turn on the “use connection settings in builder only” option, fill in the connection information so you have a live connection, then fill in the SelectCmd and use the builder to build the rest of the properties for you.

Now, anytime you need records from the Northwind Customers table, you simply use the CustomersCursor class. Of course, we haven’t defined any connection information, but that’s actually a good thing, since this class shouldn’t have to worry about things like how to get the data (ODBC, ADO, or XML) or even what database engine to use (there are Northwind databases for SQL Server, Access, and, new in version 8, VFP).



However, notice that this cursor deals with all records in the Customers table. Sometimes, you only want a specific customer. So, let's create a subclass of CustomersCursor called CustomerByIDCursor. Change SelectCmd to "select \* from customers where customerid = ?pcustomerid" and put the following code into Init:

```
lparameters tcCustomerID
dodefault()
This.AddParameter('pCustomerID', tcCustomerID)
```

This creates a parameter called pCustomerID (which is the same name as the one specified in SelectCmd) and sets it to any value passed. If no value is passed, use GetParameter to return an object for this parameter and set its Value property before calling GetData.

Create an OrdersCursor class similar to CustomersCursor except that it retrieves all records from the Orders table. Then create an OrdersForCustomerCursor subclass that retrieves only those orders for a specific customer. Set SelectCmd to "select \* from orders where customerid = ?pcustomerid" and put the same code into Init as CustomerByIDCursor has (since it's the same parameter).

To test how this works, run TestCustomersCursor.prg.

## Example: Form

Now that we have some reusable data classes, let's put them to use. First, let's create a subclass of SFDataEnvironment called CustomersAndOrdersDataEnvironment that contains CustomerByIDCursor and OrdersForCustomerCursor classes. Set AutoOpenTables to .F. (because we need to set connection information before the tables can be opened) and UseDEDDataSource for both CursorAdapters to .T. This DataEnvironment can now be used in a form to show information about a specific customer, including its orders.

Let's create such a form. Create a form called CustomerOrders.scx (it's included with the sample files accompanying this document), set DEClass and DEClassLibrary to CustomersAndOrdersDataEnvironment so we use our reusable DataEnvironment. Put the following code into the Load method:

```
#define ccDATASOURCETYPE 'ADO'
with This.CustomersAndOrdersDataEnvironment

* Set the DataEnvironment data source.

.DataSourceType = ccDATASOURCETYPE

* If we're using ODBC or ADO, create a connection manager and open a
* connection to the Northwind database.

if .DataSourceType $ 'ADO, ODBC'
    This.AddProperty('oConnMgr')
    This.oConnMgr = newobject('SFConnectionMgr' + ccDATASOURCETYPE, ;
        'SFRemote')
    with This.oConnMgr
        .cDriver = iif(ccDATASOURCETYPE = 'ADO', 'SQLOLEDB.1', ;
```

```

        'SQL Server')
        .cServer = '(local)'
        .cDatabase = 'Northwind'
        .cUserName = 'sa'
        .cPassword = ''
    endwhile
    if not This.oConnMgr.Connect()
        messagebox(oConnMgr.cErrorMessage)
        return .F.
    endif not This.oConnMgr.Connect()

* If we're using ADO, each cursor must have its own datasource.

    if .DataSourceType = 'ADO'
        .CustomerByIDCursor.UseDEDataSource = .F.
        .CustomerByIDCursor.DataSourceType = 'ADO'
        .OrdersForCustomerCursor.UseDEDataSource = .F.
        .OrdersForCustomerCursor.DataSourceType = 'ADO'
    endif .DataSourceType = 'ADO'

* Set the DataSource to the connection.

        .SetConnection(This.oConnMgr.GetConnection())

* If we're using XML, change the SelectCmd to call the GetNWXML function
* instead.

else
    .CustomerByIDCursor.SelectCmd = 'GetNWXML([customersbyid.xml?' + ;
        'customerid=] + pCustomerID) '
    .CustomerByIDCursor.UpdateCmdDataSourceType = 'XML'
    .CustomerByIDCursor.UpdateCmd = 'SendNWXML(This.UpdateGram) '
    .OrdersForCustomerCursor.SelectCmd = 'GetNWXML([ordersforcustomer.' + ;
        'xml?customerid=] + pCustomerID) '
    .OrdersForCustomerCursor.UpdateCmdDataSourceType = 'XML'
    .OrdersForCustomerCursor.UpdateCmd = 'SendNWXML(This.UpdateGram) '
endif .DataSourceType $ 'ADO,ODBC'

* Specify that the value for the cursor parameters will be filled from the
* CustomerID textbox.

loParameter = .CustomerByIDCursor.GetParameter('pCustomerID')
loParameter.Value = '=Thisform.txtCustomerID.Value'
loParameter = .OrdersForCustomerCursor.GetParameter('pCustomerID')
loParameter.Value = '=Thisform.txtCustomerID.Value'

* Create empty cursors and display an error message if we failed.

if not .GetData(.T.)
    messagebox(.cErrorMessage)
    return .F.
endif not .GetData(.T.)
endwith

```

This looks like a lot of code, but most of it is there for demo purposes to allow switching to different data access mechanisms.

This code creates a connection manager to handle the connection, either ADO, ODBC, or XML, depending on the `ccDATASOURCETYPE` constant, which you can change to try each of the mechanisms. In the case of ADO, since each `CursorAdapter` must have its own `DataSource`, the `UseDEDataSource` and `DataSourceType` properties are set for each one. The code then calls the `SetConnection` method to set up the connection information. In the case of XML, the `SelectCmd`, `UpdateCmdDataSourceType`, and `UpdateCmd` properties must be changed as discussed earlier. Next, the code uses the `GetParameter` method of both `CursorAdapter` objects to set the `Value` of the `pCustomerID` parameter to the contents of a textbox in the form. Note the use of the “=” in the value; this means the `Value` property will be evaluated every time it’s needed, so we essentially have a dynamic parameter (saving the need to constantly change the parameter to the current value as the user types in the textbox). Finally, the `GetData` method is called to create empty cursors so the data binding of the controls will work.

Drop a textbox on the form and name it `txtCustomerID`. Put the following code into its `Valid` method:

```
with Thisform
    .CustomersAndOrdersDataEnvironment.Requery()
    .Refresh()
endwith
```

This causes the cursors to be requeried and the controls to be refreshed when a customer ID is entered.

Drop a label on the form, put it beside the textbox, and set its `Caption` to “Customer ID”.

Drag the `CompanyName`, `ContactName`, `Address`, `City`, `Region`, `PostalCode`, and `Country` fields from the `Customers` cursor in the `DataEnvironment` to the form to create controls for these fields. Then select the `OrderID`, `EmployeeID`, `OrderDate`, `RequiredDate`, `ShippedDate`, `ShipVia`, and `Freight` fields in the `Orders` cursor and drag them to the form to create a grid.

That’s it. Run the form and enter “ALFKI” for the customer ID. When you tab out of the textbox, you should see the customer address information and orders appear. Try changing something about the customer or order, then closing the form, running it again, and entering “ALFKI” again. You should see that the changes you made were written to the backend database without any effort on your part.

Cool, huh? That wasn’t a lot more work than creating a form based on local tables or views. Even better, try changing the `ccDATASOURCETYPE` constant to “ADO” or “XML” and notice that the form looks and works exactly the same. That’s the whole point of `CursorAdapters`!

## Example: Report

Let’s try a report. The example discussed here was taken from `CustomerOrders.frx` that accompanies this document. The biggest issue here is that, unlike a form, we can’t tell a report to use a `DataEnvironment` subclass, nor can we drop `CursorAdapter` subclasses in the `DataEnvironment`. So, we’ll have to put some code into the report to add `CursorAdapter` subclasses to the `DataEnvironment`. Although it might seem logical to put this code into the

BeforeOpenTables event of the report's DataEnvironment, that won't actually work because for reasons I don't understand, BeforeOpenTables fires on every page when you preview the report. So, we'll put the code into the Init method.

```
#define ccDATASOURCETYPE 'ODBC'
with This
    set safety off

* Set the DataEnvironment data source.

    .DataSourceType = ccDATASOURCETYPE

* Create CursorAdapter objects for Customers and Orders.

    .NewObject('CustomersCursor', 'CustomersCursor', 'NorthwindDataClasses')
    .CustomersCursor.AddTag('CustomerID', 'CustomerID')
    .NewObject('OrdersCursor', 'OrdersCursor', 'NorthwindDataClasses')
    .OrdersCursor.AddTag('CustomerID', 'CustomerID')

* If we're using ODBC or ADO, create a connection manager and open a
* connection to the Northwind database.

if .DataSourceType $ 'ADO,ODBC'
    .AddProperty('oConnMgr')
    .oConnMgr = newobject('SFConnectionMgr' + ccDATASOURCETYPE, ;
        'SFRemote')
    with .oConnMgr
        .cDriver = iif(ccDATASOURCETYPE = 'ADO', 'SQLOLEDB.1', ;
            'SQL Server')
        .cServer = '(local)'
        .cDatabase = 'Northwind'
        .cUserName = 'sa'
        .cPassword = ''
    endwith
    if not .oConnMgr.Connect()
        messagebox(.oConnMgr.cErrorMessage)
        return .F.
    endif not .oConnMgr.Connect()

* If we're using ADO, each cursor must have its own datasource.

if .DataSourceType = 'ADO'
    .CustomersCursor.UseDEDataSource = .F.
    .CustomersCursor.DataSourceType = 'ADO'
    .CustomersCursor.SetConnection(.oConnMgr.GetConnection())
    .OrdersCursor.UseDEDataSource = .F.
    .OrdersCursor.DataSourceType = 'ADO'
    .OrdersCursor.SetConnection(.oConnMgr.GetConnection())
else
    .CustomersCursor.UseDEDataSource = .T.
    .OrdersCursor.UseDEDataSource = .T.
    .DataSource = .oConnMgr.GetConnection()
endif .DataSourceType = 'ADO'
.CustomersCursor.SetConnection(.oConnMgr.GetConnection())
.OrdersCursor.SetConnection(.oConnMgr.GetConnection())
```

```

* If we're using XML, change the SelectCmd to call the GetNWXML function
* instead.

else
    .CustomersCursor.SelectCmd      = 'GetNWXML([getallcustomers.xml]) '
    .CustomersCursor.DataSourceType = 'XML'
    .OrdersCursor.SelectCmd        = 'GetNWXML([getallorders.xml]) '
    .OrdersCursor.DataSourceType   = 'XML'
endif .DataSourceType $ 'ADO,ODBC'

* Get the data and display an error message if we failed.

if not .CustomersCursor.GetData()
    messagebox(.CustomersCursor.cErrorMessage)
    return .F.
endif not .CustomersCursor.GetData()
if not .OrdersCursor.GetData()
    messagebox(.OrdersCursor.cErrorMessage)
    return .F.
endif not .OrdersCursor.GetData()

* Set a relation from Customers to Orders.


set relation to CustomerID into Customers
endwith

```

This code looks similar to that of the form. Again, the majority of the code is to handle different data access mechanisms. However, there is some additional code because we can't use a `DataEnvironment` subclass and have to code the behavior ourselves.

Now, how do we conveniently put the fields on the report? Since the `CursorAdapter` don't exist in the `DataEnvironment` at design time, we can't just drag fields from them to the report. Here's a tip: create a PRG that creates the cursors and leaves them in scope (either by suspending or making the `CursorAdapter` objects public), then use the Quick Report function to put fields with the proper sizes on the report.

Create a group on `CUSTOMERS.CUSTOMERID` and check "Start each group on a new page". Then lay out the report to look similar to the following:



## XMLAdapter

In addition to CursorAdapter, VFP 8 has three new base classes to improve VFP's support for XML: XMLAdapter, XMLTable, and XMLField. XMLAdapter provides a means of converting data between XML and VFP cursors. It has a lot more capabilities than the CURSORTOXML() and XMLTOCURSOR() functions, including support for hierarchical XML and the ability to consume types of XML those function don't support such as ADO.NET DataSets. XMLTable and XMLField are child objects that provide the ability to fine-tune the schema for the XML data. In addition, XMLTable has an ApplyDiffgram method intended to allow VFP to consume updategrams and diffgrams, something that was missing from VFP 7.

To give you an idea of its capabilities, I created an ASP.NET Web Service that returns an ADO.NET DataSet, and then used an XMLAdapter object in VFP to consume that DataSet. Here's now I did this.

First, in Visual Studio.NET, I dragged the Northwind Customers table from the Server Explorer to a new ASP.NET Web Service project I called NWWebService. This automatically created two objects, SqlConnection1 and SqlDataAdapter1. I then added the following code to the existing generated code:

```
<WebMethod()> Public Function GetAllCustomers() As DataSet
    Dim loDataSet As New DataSet()
    Me.SqlConnection1.Open()
    Me.SqlDataAdapter1.Fill(loDataSet)
    Return loDataSet
End Function
```

I built the project to generate the appropriate Web Service files in the NWWebService virtual directory (which VS.NET automatically created for me).

To consume this Web Service in VFP, I used the IntelliSense Manager to register a Web Service called "Northwind.NET", pointing it at "http://localhost/NWWebService/NWWebService.asmx?WSDL" as the location of the WSDL file. I then created the following code (in XMLAdapterWebService.prg) to call the Web Service and convert the ADO.NET DataSet into a VFP cursor.

```
local loWS as Northwind.NET, ;
    loXMLAdapter as XMLAdapter, ;
    loTable as XMLTable

* Get the .NET DataSet from a .NET Web Service.

loWS = NEWOBJECT("Wsclient",HOME()+"ffc\_webservices.vcx")
loWS.cWSName = "Northwind.NET"
loWS = loWS.SetupClient("http://localhost/NWWebService/NWWebService.asmx" + ;
    "?WSDL", "NWWebService", "NWWebServiceSoap")
loXML = loWS.GetAllCustomers()

* Create an XMLAdapter and load the data.

loXMLAdapter = createobject('XMLAdapter')
loXMLAdapter.XMLSchemaLocation = '1'
loXMLAdapter.LoadXML(loXML.Item(0).parentnode.xml)

* If we succeeded in loading the XML, create and browse a cursor from each
* table object.

if loXMLAdapter.IsLoaded
    for each loTable in loXMLAdapter.Tables
        loTable.ToCursor()
        browse
        use
    next loTable
endif loXMLAdapter.IsLoaded
```


Note that in order to use XMLAdapter, you need MSXML 4.0 Service Pack 1 or later installed on your system. You can download this from the MSDN Web site (go to <http://msdn.microsoft.com> and search for MSXML).

## Summary

I think CursorAdapter is one of the biggest and most exciting enhancements in VFP 8 because it provides a consistent and easy-to-use interface to remote data, plus it allows us to create reusable data classes. I'm sure once you work with them, you'll find them as exciting as I do.

## Biography

Doug Hennig is a partner with Stonefield Systems Group Inc. He is the author of the award-winning Stonefield Database Toolkit (SDT) and co-author of the award-winning Stonefield Query. He is co-author (along with Tamar Granor, Ted Roche, and Della Martin) of "The Hacker's Guide to Visual FoxPro 7.0" and co-author (along with Tamar Granor and Kevin McNeish) of "What's New in Visual FoxPro 7.0", both from Hentzenwerke Publishing, and author of "The Visual FoxPro Data Dictionary" in Pinnacle Publishing's Pros Talk Visual FoxPro series. He writes the monthly "Reusable Tools" column in FoxTalk. He was the technical editor of "The Hacker's Guide to Visual FoxPro 6.0" and "The Fundamentals", both from Hentzenwerke Publishing. Doug has spoken at every Microsoft FoxPro Developers Conference (DevCon) since 1997 and at user groups and developer conferences all over North America. He is a Microsoft Most Valuable Professional (MVP) and Certified Professional (MCP).



**stonefield**   
S Y S T E M S   G R O U P   I N C

Copyright © 2002 Doug Hennig. All Rights Reserved


Doug Hennig  
Partner  
Stonefield Systems Group Inc.  
1112 Winnipeg Street, Suite 200  
Regina, SK Canada S4R 1J6  
Phone: (306) 586-3341 Fax: (306) 586-5080  
Email: [dhennig@stonefield.com](mailto:dhennig@stonefield.com)  
Web: [www.stonefield.com](http://www.stonefield.com)



## Appendix: Setting Up SQL Server 2000 XML Access

In order to access SQL Server 2000 using a URL in a browser or other HTTP client, you have to do a few things. First, you need to download and install SQLXML 3.0 from the MSDN Web site (<http://msdn.microsoft.com>; do a search for “SQLXML” and then choose the download link).

Next, you have to set up an IIS virtual directory. To do this, choose the Configure IIS Support shortcut in the SQLXML 3.0 folder under Start, Programs in your Windows Taskbar. Expand the node for your server, choose the Web site to work with, and then right-click and choose New, Virtual Directory. In the General tab of the dialog that appears, enter the name of the virtual directory and its physical path. For the samples in this document, use “Northwind” as the virtual directory name and “NorthwindTemplates” for the physical directory. Using Windows Explorer, create the physical directory somewhere on your system, and create a subdirectory of it called “Template” (we’ll use that subdirectory in a moment). Copy the templates files included with this article to the Template subdirectory.




In the Security tab, enter the appropriate information to access SQL Server, such as the user name and password or the specific authentication mechanism you want to use.

The image shows a screenshot of the "Northwind Properties" dialog box, specifically the "Security" tab. The dialog box has a title bar with a question mark and a close button. Below the title bar, there are four tabs: "Virtual Names", "Advanced", "Upgrade to version 3", and "Security". The "Security" tab is selected. The main area of the dialog box contains the following elements:


- A heading: "Select the authentication method you want to use when users access data."
- A radio button labeled "Always log on as:" which is selected.
- A "Credentials:" section containing:
  - A "User Name:" label followed by a text box containing "sa".
  - A "Password:" label followed by a password box containing asterisks.
  - A "Confirm Password:" label followed by an empty password box.
  - An "Account Type:" label with two radio buttons: "SQL Server" (selected) and "Windows".
  - A checkbox labeled "Enable Windows account synchronization" which is unchecked.
- Two other radio buttons: "Use Windows Integrated Authentication" and "Use Basic Authentication (Clear Text) to SQL Server account", both of which are unselected.

At the bottom of the dialog box, there are three buttons: "OK", "Cancel", and "Apply".


In the Data Source tab, choose the server and, if desired, the database to use. For the examples for this document, choose the Northwind database.




In the Settings tab, choose the desired settings, but at a minimum, turn on Allow Template Queries and Allow POST.



In the Virtual Names tab, choose “template” from the Type combobox and enter a virtual name (for this document, use “template”) and physical path (which should be a subdirectory of the virtual directory; for this document, use “Template”) to use for templates. Click on OK



To test this, bring up your browser and type the following URL:  
<http://localhost/northwind/template/getallcustomers.xml>. You should see something like the following:



```
- <root xmlns:sql="urn:schemas-microsoft-com:xml-sql" >
  <Customers CustomerID="ALFKI" CompanyName="Alfreds Futterkiste"
    ContactName="Maria Anders" ContactTitle="Sales Representative"
    Address="Obere Str. 57" City="Berlin" PostalCode="12209" Country="Germany"
    Phone="030-0074321" Fax="999-999-9999" />
  <Customers CustomerID="ANATR" CompanyName="Ana Trujillo Emparedados y
    helados" ContactName="Ana Trujillo" ContactTitle="Owner" Address="Avda. de la
    Constitución 2222" City="México D.F." PostalCode="05021" Country="Mexico"
    Phone="(5) 555-4729" Fax="(5) 555-3745" />
  <Customers CustomerID="ANTON" CompanyName="Antonio Moreno Taquería"
    ContactName="Antonio Moreno" ContactTitle="Owner" Address="Mataderos
    2312" City="México D.F." PostalCode="05023" Country="Mexico" Phone="(5)
    555-3932" />
  <Customers CustomerID="AROUT" CompanyName="Around the Horn"
    ContactName="Thomas Hardy" ContactTitle="Sales Representative"
    Address="120 Hanover Sq." City="London" PostalCode="WA1 1DP"
    Country="UK" Phone="(171) 555-7788" Fax="(171) 555-6750" />
  <Customers CustomerID="BERGS" CompanyName="Berglunds snabbköp"
    ContactName="Christina Berglund" ContactTitle="Order Administrator"
    Address="Berguvsvägen 8" City="Luleå" PostalCode="S-958 22"
    Country="Sweden" Phone="0921-12 34 65" Fax="0921-12 34 67" />
  <Customers CustomerID="BLAUS" CompanyName="Blauer See Delikatessen"
    ContactName="Hanna Moos" ContactTitle="Sales Representative"
    Address="Forsterstr. 57" City="Mannheim" PostalCode="68306"
    Phone="0921-3643 10" Fax="0921-3643 12" />
```